Křivkový integrál skalárního pole

V následujících příkladech budeme počítat křivkové integrály skalárního pole f podél křivky K. Příkazy jsou zapsány červenou barvou, "maplovská" řešení pak barvou modrou.

Příkazy, které budeme potřebovat, jsou obsaženy v balíku programů `VectorCalculus`, který si otevřeme následujícím příkazem :

```
> with(VectorCalculus);
```

```
Warning, the assigned names <,> and <|> now have a global binding

Warning, these protected names have been redefined and
unprotected: *, +, ., D, Vector, diff, int, limit, series
```

$$[\&x, *, +, ., <,>, <|>, \mathit{AddCoordinates}, \mathit{ArcLength}, \mathit{BasisFormat}, \mathit{Binormal}, \mathit{CrossProduct}, \mathit{Curl}, \mathit{Curvature}, \mathrm{D}, \mathit{Del}, \mathit{DirectionalDiff}, \mathit{Divergence}, \mathit{DotProduct}, \mathit{Flux}, \mathit{GetCoordinateParameters}, \mathit{GetCoordinates}, \mathit{Gradient}, \mathit{Hessian}, \mathit{Jacobian}, \mathit{Laplacian}, \mathit{LineInt}, \mathit{MapToBasis}, \mathit{Nabla}, \mathit{PathInt}, \mathit{PrincipalNormal}, \mathit{RadiusOfCurvature}, \mathit{ScalarPotential}, \mathit{SetCoordinateParameters}, \mathit{SetCoordinates}, \mathit{SurfaceInt}, \mathit{TNBFrame}, \mathit{Tangent}, \mathit{TangentLine}, \mathit{TangentPlane}, \mathit{TangentVector}, \mathit{Torsion}, \mathit{Vector}, \mathit{VectorField}, \mathit{VectorPotential}, \mathit{Wronskian}, \mathit{diff}, \mathit{evalVF}, \mathit{int}, \mathit{limit}, \mathit{series}]$$

V prvních 9 příkladech budeme integrovat skalární pole $f : R^2 \to R$, a to podél 2-rozměrné křivky. Většinou budeme pracovat s kartézskými souřadnicemi. Musíme to však "Maplu sdělit". To se provede pomocí následujícího příkazu:

```
> SetCoordinates(cartesian[x,y]);
```

$$cartesian_{x,y}$$

Příklad 1: $f(x,y) = xy$, K je úsečka PQ, $P = (1,0)$, $Q = (0,2)$.
Řešení :

```
> PathInt(x*y,[x,y]=Line(<1,0>,<0,2>),'inert')=
> PathInt( x*y,[x,y] = Line( <1,0>, <0,2> ));
```

$$\int_0^1 2\,(1-t)\,t\,\sqrt{5}\,dt = \frac{\sqrt{5}}{3}$$

Příklad 2: $f(x,y) = x^2 + y^2$, K je úsečka PQ, $P = (1,1)$, $Q = (3,3)$.
Řešení :

```
> PathInt( x^2+y^2, [x,y] = Line( <1,1>, <3,3> ),'inert' )=
> PathInt(x^2+y^2, [x,y] = Line( <1,1>, <3,3> ));
```

$$\int_0^1 4\,(1+2\,t)^2\,\sqrt{2}\,dt = \frac{52\,\sqrt{2}}{3}$$

Příklad 3: $f(x,y) = x + y$, K je obvod trojúhelníka s vrcholy $A = (0,0)$, $B = (0,2)$ $C = (1,0)$.
Řešení :
Protože na křivku lze v tomto případě nahlížet také jako na lomenou čáru, která začíná a končí ve stejném bodě, lze křivkový integrál spočítat pomocí následujícího příkazu:

```
> PathInt(x+y,[x,y]=LineSegments(<0,0>,<0,2>,<1,0>,<0,0>),'inert')=
> PathInt(x+y,[x,y]=LineSegments( <0,0>, <0,2>,<1,0>,<0,0> ));
```

$$\int_0^1 4\,t\,dt + \int_0^1 (-t+2)\sqrt{5}\,dt + \int_0^1 1-t\,dt = \frac{5}{2} + \frac{3\sqrt{5}}{2}$$

Příklad 4: $f(x,y) = x^2,\ K$ je oblouk křivky $y = \ln x,\ \ x \in <1,2>$.

Řešení :

Křivku K lze parametrizovat takto: $x = t,\ y = \ln t,\ t \in <1,2>$.

Křivkový integrál potom spočteme pomocí násl. příkazu:

```
> PathInt( x^2, [x,y] = Path( <t,ln(t)>, t=1..2 ),'inert' )=
> PathInt(x^2, [x,y] = Path( <t,ln(t)>, t=1..2 ));
```

$$\int_1^2 t^2\sqrt{1+\frac{1}{t^2}}\,dt = -\frac{2\sqrt{2}}{3} + \frac{5\sqrt{5}}{3}$$

Příklad 5: $f(x,y) = x,\ K$ je oblouk paraboly $y = x^2$, s koncovými body $A = (1,1),\ B = (2,4)$.

Řešení :

Křivku K lze parametrizovat takto: $x = t,\ y = t^2,\ t \in <1,2>$.

Křivkový integrál potom spočteme pomocí násl. příkazu:

```
> PathInt( x, [x,y] = Path( <t,t^2>, t=1..2 ),'inert' )=
> PathInt( x,[x,y] = Path( <t,t^2>, t=1..2 ));
```

$$\int_1^2 t\sqrt{1+4\,t^2}\,dt = -\frac{5\sqrt{5}}{12} + \frac{17\sqrt{17}}{12}$$

Příklad 6: $f(x,y) = y,\ K$ je oblouk paraboly $y^2 = 2px$, s koncovými body $A = (0,0),\ B = (0,2)$.

Řešení :

Křivku K lze parametrizovat takto: $x = \frac{t^2}{2p},\ y = t,\ t \in <0,2>$.

Křivkový integrál potom spočteme pomocí násl. příkazu:

```
> PathInt( y, [x,y] = Path( <t^2/(2*p),t>, t=0..2 ),'inert' )=
> PathInt(y, [x,y] = Path( <t^2/(2*p),t>, t=0..2 ));
```

$$\int_0^2 t\,\sqrt{1+\frac{t^2}{p^2}}\,dt = -\frac{p^2}{3} + \frac{(\frac{p^2+4}{p^2})^{(3/2)}\,p^2}{3}$$

Příklad 7: $f(x,y) = \sqrt{2y}$, K je větev cykloidy zadané parametricky: $x = 2(t-\sin t)$, $y = 2(1-\cos t)$, $t \in < 0, 2\pi >$

Řešení :

```
> PathInt( sqrt(2*y), [x,y] = Path( <2*(t-sin(t)),2*(1-cos(t))>,
> t=0..2*Pi ),'inert' )=PathInt( sqrt(2*y), [x,y] =
> Path(<2*(t-sin(t)),2*(1-cos(t))>, t=0..2*Pi ));
```

$$\int_0^{2\,\pi} \sqrt{2}\,\sqrt{-2\cos(t)+2}\,\sqrt{(-2\cos(t)+2)^2+4\sin(t)^2}\,dt = 8\,\sqrt{2}\,\pi$$

Příklad 8: $f(x,y) = \operatorname{arctg}\frac{y}{x}$, K je část Archimedovy spirály zadané v polárních souřadnicích rovnicí $r = \varphi$, $\varphi \in < 0, 1 >$. **Řešení :**

1.způsob: Křivku lze zadat jako "maplovský" `Vector` v polárních souřadnicích: $< r(\varphi), \varphi > = < \varphi, \varphi > = < t, t >$, přejmenujeme-li proměnnou φ na t.

Křivkový integrál potom spočteme pomocí násl. příkazu:

```
> PathInt( arctan(y/x), [x,y] = Path( <t,t>, t=0..1,'coords'='polar'
> ),'inert' )=PathInt( arctan(y/x), [x,y] = Path( <t,t>,
> t=0..1,'coords'='polar' ));
```

$$\int_0^1 \arctan(\frac{\sin(t)}{\cos(t)})\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt =$$
$$\int_0^1 \arctan(\frac{\sin(t)}{\cos(t)})\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt$$

Vidíme, že Maple integrál nespočítal. Zkusme tedy

2.způsob: Ten spočívá v tom, že pomocí transformačních vztahů mezi polárními a kartézskými souřadnicemi sestavíme parametrické rovnice křivky K v kartézských souřadnicích:

$x = r\cos t = t\cos t,\quad y = r\sin t = t\sin t,\quad t \in < 0, 1 >$.

Křivkový integrál potom spočteme pomocí násl. příkazu:

```
> PathInt( arctan(y/x), [x,y] = Path( <t*cos(t),t*sin(t)>, t=0..1
> ),'inert' )=PathInt( arctan(y/x), [x,y] = Path( <t*cos(t),t*sin(t)>,
> t=0..1 ));
```

$$\int_0^1 \arctan(\frac{\sin(t)}{\cos(t)})\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt =$$
$$\int_0^1 \arctan(\frac{\sin(t)}{\cos(t)})\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt$$

Vidíme, že Maple integrál opět nespočítal. Zkusme "mu tedy pomoci" úpravou integrandu:

```
> Int(arctan(tan(t))*sqrt((cos(t)-t*sin(t))^2+(sin(t)+t*cos(t))^2),t=0.
> .1)=int(arctan(tan(t))*sqrt((cos(t)-t*sin(t))^2+(sin(t)+t*cos(t))^2),t
> =0..1);
```

$$\int_0^1 \arctan(\tan(t))\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt =$$

$$\int_0^1 \arctan(\tan(t))\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt$$

Maple integrál znovu nespočítal. Zkusme "mu tedy pomoci" ještě následující úpravou integrandu:

$$\operatorname{arctg}\operatorname{tg} t = t, \quad \text{protože} \;\; t \in < 0,1 > \subset < -\frac{\pi}{2}, \frac{\pi}{2} > .$$

```
> Int(t*sqrt((cos(t)-t*sin(t))^2+(sin(t)+t*cos(t))^2),t=0..1)=int(t*sqr
> t((cos(t)-t*sin(t))^2+(sin(t)+t*cos(t))^2),t=0..1);
```

$$\int_0^1 t\,\sqrt{(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt = \frac{2\sqrt{2}}{3}-\frac{1}{3}$$

Příklad 9: $f(x,y) = \frac{1}{(x^2+y^2)^{\frac{3}{2}}}$, K je část hyperbolické spirály zadané v polárních souřadnicích rovnicí $r = \frac{1}{\varphi}$, $\varphi \in < \sqrt{3}, 2\sqrt{2} >$.

Řešení :

Z hlediska způsobu řešení máme 2 možnosti-stejné jako v předchozím příkladě.

1.způsob:

```
> PathInt( 1/(x^2+y^2)^(3/2), [x,y] = Path( <1/t,t>,
> t=sqrt(3)..2*sqrt(2),'coords'='polar' ),'inert' )=PathInt(
> 1/(x^2+y^2)^(3/2), [x,y] = Path( <1/t,t>,
> t=sqrt(3)..2*sqrt(2),'coords'='polar' ));
```

$$\int_{\sqrt{3}}^{2\sqrt{2}} \frac{\sqrt{(-\frac{\cos(t)}{t^2} - \frac{\sin(t)}{t})^2 + (-\frac{\sin(t)}{t^2} + \frac{\cos(t)}{t})^2}}{(\frac{\cos(t)^2}{t^2} + \frac{\sin(t)^2}{t^2})^{(3/2)}}\, dt = \frac{19}{3}$$

2.způsob:

```
> PathInt( 1/(x^2+y^2)^(3/2), [x,y] = Path( <1/t*cos(t),1/t*sin(t)>,
> t=sqrt(3)..2*sqrt(2) ),'inert' )=PathInt( 1/(x^2+y^2)^(3/2), [x,y] =
> Path( <1/t*cos(t),1/t*sin(t)>, t=sqrt(3)..2*sqrt(2) ));
```

$$\int_{\sqrt{3}}^{2\sqrt{2}} \frac{\sqrt{(-\frac{\cos(t)}{t^2} - \frac{\sin(t)}{t})^2 + (-\frac{\sin(t)}{t^2} + \frac{\cos(t)}{t})^2}}{(\frac{\cos(t)^2}{t^2} + \frac{\sin(t)^2}{t^2})^{(3/2)}}\, dt = \frac{19}{3}$$

V následujících 3 příkladech budeme integrovat skalární pole $f : R^3 \to R$, a to podél 3-rozměrné křivky. Protože budeme pracovat s 3-rozměrnými kartézskými souřadnicemi, musíme si je umět v Maplu "vyvolat". To se provede pomocí násl. příkazu:

```
> SetCoordinates(cartesian[x,y,z]);
```

$$cartesian_{x,\,y,\,z}$$

Příklad 10: $f(x, y, z) = \frac{z^2}{(x^2+y^2)}$, K je závit šroubovice $x = 2\cos t,\ y = 2\sin t,\ z = 2t,\ t \in <0, 2\pi>$.

Řešení :

```
> PathInt( z^2/(x^2+y^2), [x,y,z] = Path( <2*cos(t),2*sin(t),2*t>,
> t=0..2*Pi ),'inert' )=PathInt( z^2/(x^2+y^2), [x,y,z] = Path(
> <2*cos(t),2*sin(t),2*t>, t=0..2*Pi ));
```

$$\int_0^{2\,\pi} \frac{8\,t^2\,\sqrt{1+\sin(t)^2+\cos(t)^2}}{4\sin(t)^2+4\cos(t)^2}\,dt = \frac{16\,\pi^3\,\sqrt{2}}{3}$$

Příklad 11: $f(x,y,z)=\frac{1}{x^2+y^2+z^2}$, K je dána parametrickými rovnicemi $x=\mathrm{e}^t\cos t,\ y=\mathrm{e}^t\sin t,\ z=\mathrm{e}^t,\ t\in<0,1>$.

Řešení :

```
> PathInt( 1/(x^2+y^2+z^2), [x,y,z] = Path(
> <exp(t)*cos(t),exp(t)*sin(t),exp(t)>, t=0..1 ),'inert' )=PathInt(
> 1/(x^2+y^2+z^2), [x,y,z] = Path( <exp(t)*cos(t),exp(t)*sin(t),exp(t)>,
> t=0..1 ));
```

$$\int_0^1 \frac{\sqrt{(e^t\cos(t)-e^t\sin(t))^2+(e^t\sin(t)+e^t\cos(t))^2+(e^t)^2}}{(e^t)^2\cos(t)^2+(e^t)^2\sin(t)^2+(e^t)^2}\,dt = \frac{\sqrt{3}}{2}-\frac{1}{2}\sqrt{3}\,e^{(-1)}$$

Příklad 12: $f(x,y,z)=z$, K je dána parametrickými rovnicemi $x=t\cos t,\ y=t\sin t,\ z=t,\ t\in<0,2\pi>$.

Řešení :

```
> PathInt(z, [x,y,z] = Path( <t*cos(t),t*sin(t),t>, t=0..2*Pi ),'inert'
> )=PathInt(z, [x,y,z] = Path( <t*cos(t),t*sin(t),t>, t=0..2*Pi ));
```

$$\int_0^{2\,\pi} t\,\sqrt{1+(\cos(t)-t\sin(t))^2+(\sin(t)+t\cos(t))^2}\,dt = -\frac{2\,\sqrt{2}}{3}+\frac{(2+4\,\pi^2)^{(3/2)}}{3}$$